'FoxCode - A Word for Windows 6.0 Macro

'

'	Original:	Dina Pavlis, Seattle, WA

'	Changes :	Jefferey Donnici, Westminster, CO

'

'	Released into the Public Domain on April 11, 1994 with 

'	no warranty whatsoever, expressed nor implied.

'

'	To install this macro, select the full text of the document

'	(CTRL-NumPad 5) and choose Edit|Copy from the menu.  Then

'	go to the tools menu and select Macro...   Enter FoxCode 

'	into the Macro Name box and select Create.  Erase the 

'	Sub MAIN and End Sub that are created and select Edit|Paste

'	from the menu.  Save the macro.  

'

'	To use the macro, load a FoxPro program (or any ASCII FoxPro

'	code file) into Word.  Select Macro... from the Tools menu

'	again.  Select the FoxCode macro and hit Run.  Sit back and

'	watch....

'

'	The code has been heavily commented to allow for any 

'	customization you might want to do.  The fonts I've used

'	might not appeal to you.	

'	

Sub MAIN



DisableInput 0	'	Allow user to press ESC key during execution



'	First, set up the page.  Wade through there and you can see the

'	margin settings and page size/



FilePageSetup .Tab = "0", .PaperSize = "0", .TopMargin = "1" + Chr$(34), .BottomMargin = "0.5" + Chr$(34), .LeftMargin = "0.5" + Chr$(34), .RightMargin = "0.5" + Chr$(34), .Gutter = "0" + Chr$(34), .PageWidth = "8.5" + Chr$(34), .PageHeight = "11" + Chr$(34), .Orientation = 0, .FirstPage = 0, .OtherPages = 0, .VertAlign = 0, .ApplyPropsTo = 4, .FacingPages = 0, .HeaderDistance = "0.5" + Chr$(34), .FooterDistance = "0.5" + Chr$(34), .SectionStart = 2, .OddAndEvenPages = 0, .DifferentFirstPage = 0, .Endnotes = 0, .LineNum = 0, .StartingNum = "", .FromText = "", .CountBy = "0", .NumMode = - 1

LineUp 1



EndOfDocument 		'	Go to the end of the document



InsertPara			'	Insert a blank line so the loop will know

				'	when to end



StartOfDocument		'	Go to the beginning of the document



ViewHeader			'	Go into the Page Header for formatting



Font "Arial", 12		'	The Page Header/Filename font.

Bold



'	Insert the filename field.  Will update with every printing.

InsertField .Field = "FILENAME \* Upper\p \* MERGEFORMAT"

InsertPara			'	Insert a blank line



Font "Arial", 10		'	Change the font size down to 10



InsertPara			'	ANOTHER blank line (I like space! <g>)



'	Insert the Date field.  Will also update with every print.  By

'	playing with the "dddd, MMMM dd, yyyy" settings, you can format

'	the way the date should appear.

InsertField .Field = "DATE \@ " + Chr$(34) + "dddd, MMMM dd, yyyy" + Chr$(34) + " \* MERGEFORMAT"



'	Insert the Time field and move it over to the right.  Again, 

'	playing with the h:mm:ss AM/PM" will change the time format.

Insert " "

InsertField .Field = "TIME \@ " + Chr$(34) + "h:mm:ss AM/PM" + Chr$(34) + " \* MERGEFORMAT"

WordLeft 6

Insert Chr$(9) + Chr$(9)

EndOfLine



InsertPara			'	Inserts a space between the Page Header

				'	and the code



GoToHeaderFooter		'	Go to the Page Footer



Insert " "

InsertPara			'	Insert a space between the code and the Page

				'	Footer.

CenterPara			'	Center the Page Number

Font "Arial", 10		'	Select the font for "Page No. x"

Bold

Insert "Page No. "



'	Inserts the current Page Number field.

InsertField .Field = "PAGE  \* MERGEFORMAT"

ViewHeader			'	Turn off the header/footer view.



'____________________________________________________________________

'



EditSelectAll		'	Select the entire document

Font "Courier New", 8	'	This is the font for the CODE portion

						'	of our program files.



StartOfDocument		'	Go to the beginning of the document



'Begin formatting lines

'_______________________



While CmpBookmarks("\Sel", "\EndOfDoc")

LoopStart:

	sChar$ = Selection$()		

	If sChar$ = "*" Then			'If it's an asterisk

		EndOfLine 1				'Select the line

		Font "Arial", 8			'Font used for COMMENT lines

		Italic 1				'Italicize it

	ElseIf sChar$ = Chr$(9) Then		'It's a tab

		CharRight 1				'Move to the next character

		Goto LoopStart :			'Restart Loop

	ElseIf sChar$ = Chr$(32) Then		'It's a space

		CharRight 1				'Move to the next character

		Goto LoopStart :			'Restart Loop

	Else						'or, if not an asterisk 

		EndOfLine 1				'Highlite to end of the line.

		Bold 1				'Bold the code

	End If

	StartOfLine					'Go to start of current line

	LineDown 1					'Move down to next line

Wend							'Loop



StartOfDocument		'Go back to the top of the doc.





'	Handling in-line code (as indicated by the occurence of "&&")

'	______________________________________________________________



'	Tell Word what to look for and how to look for it.

EditFind .Find = "&&", .WholeWord = 0, .Direction = 0, .Format = 0



While EditFindFound()		'As long as && is found, keep looping

	EndOfLine 1			'When found, select the line

	Font "Arial", 8		'Change the font to the "comment" font.

	Italic 1			'Italicize it

	Bold 0			'Turn off the bold that was turned on

					'by this being a "code" line.

	EndOfLine			'Move to the end of the line occurence



	'	Restart the search

	EditFind .Find = "&&", .WholeWord = 0, .Direction = 0, .Format = 0

Wend



StartOfDocument			'Return to top of document



'	Place a message on the status bar...

Print "The formatting is complete..."



End Sub



